



APPLICATION NOTE

1

Using the Garmin GPS-16-HVS & GPS-17-HVS with the Sutron 9210 & Xpert Data Loggers

Garmin GPS-16-HVS & GPS-17-HVS Used With Sutron's 9210 & Xpert Data Loggers



Prepared by:
R&D
September, 2005

Sutron Corporation
21300 Ridgetop Circle
Sterling, Virginia 20166
TEL: (703) 406-2800
FAX: (703) 406-2801
WEB: <http://www.sutron.com/>



INTRODUCTION

The Garmin GPS-16-HVS GPS receiver (available from Sutron as part 5000-0175) and the GPS-17-HVS (5000-0176) have RS232 interfaces compatible with the Sutron 9210/Xpert. An Xpert Basic program has been written with a scheduled routine (GARMINGPS) to read the time from the GPS unit and set the time in the 9210/Xpert. The GARMINGPS routine is written to read a GPS connected to COM2 to read a NEMA GPGLGA message at 4800 baud. The factory default configuration for the Garmin GPS-16 HVS includes the GPGLGA message at 4800 baud so no additional configuration is normally needed. Other NEMA messages may be output; however, the BASIC program will ignore them and only accept the GPGLGA message.

Once the GARMINGPS.BAS program has been loaded in to the system, routine must be scheduled using Setup, Basic. Sutron recommends that the program be scheduled to run every 6 hours. This can be changed to any interval desired in order to conserve power and still maintain time.

The GARMIN GPS is wired to COM2 as follows.

Note: Sutron also offers a RJ45 to DB9 adapter that implements the required wiring.

RJ45	DB9 Female
1 PWR + (Blue)	7
2 PWR GND (Orange)	5
3 On/Off (Black)	-- use a DIO to control this or connect to GND to remain on always
4 Data In (Red)	3
5 Data Out (Grn)	2

Note that the GARMIN GPS gets power (+12V) from the DB9 pin 7. This will happen only if the COM2 jumper inside the 9210 has been set between pins 4 and 6. For the Xpert, the jumpers are J1:2-3, J2:1-2. Be sure to open the 9210/Xpert and set the jumper appropriately.

Note also that the GARMIN GPS can be turned on and off using the RJ45 pin 3 black wire. The code is written to turn on and off the GPS using DIO 4 (Terminal B6). This is done to help conserve power since it is only necessary to read the GPS a few times a day.

You can interact directly with the GARMIN GPS through the 9210 for troubleshooting using the new Passthru command available at the command line prompt. Stop recording and then issue the command Passthru com2:4800.

The GPGLGA message is an ASCII message with values separated by commas. The program uses the "WAITFOR" function to pass by all the other data that may be output from the GPS and wait for the GPGLGA message. The Input statement is then to read all the values out into the appropriate variables. The program then checks to see if the GPS unit has satellites in view and a valid time. If not, it loops until this happens or the loop times out after 900 seconds.

With a valid time, the code will only set the time if it is more than 1 second different from the 9210 time. Note that the UTC time is used, not local time. Additional checks to the time can be added if desired.



The format of the GPGGA message is

\$GPGGA,<1>,<2>,<3>,<4>,<5>,<6>,<7>,<8>,<9>,M,<10>,<11>,<12>,*hh<CR><LF>

where:

<1> UTC time of position fix, hhmmss format
<2> Latitude, ddmm.mmmmm format (leading zeros will be transmitted)
<3> Latitude hemisphere, N or S
<4> Longitude, dddmm.mmmmm format (leading zeros will be transmitted)
<5> Longitude hemisphere, E or W
<6> GPS quality indication, 0=fix not available, 1= non-differential GPS fix available, 2=differential GPS fix available, 6=estimated
<7> Number of satellites in use, 00 to 12 (leading zeros will be transmitted)
<8> Horizontal dilution of precision, 0.5 to 99.9
<9> Antenna height above/below mean sea level, -9999.9 to 99999.9 meters
<10> Geoidal height, -999.9 to 9999.9 meters
<11> Differential GPS data age
<12> Differential Reference Station ID, 0000 to 1023.
*hh checksum (XOR sum of all characters between \$ and *, non-inclusive.

The details of the \$GPGGA and the complete operations the Garmin GPS-16 are provided in a separate manual available from Garmin.

GARMINGPS.BAS

```
Public Sub SCHED_GARMINGPS
' this subroutine runs periodically (e.g. every hour) to
' read the GPS and set the Xpert clock.
' the routine turns on the GPS using Output 4
' the routine expects the GPS to be connected to COM2
' the routine expects the data in NEMA $GPGGA format
Const BAUD = 4800
Const NOPARITY = 0
Const NOHANDSHAKE = 0
Const GPSTimeout = 900 'allow 60*15minutes to get the GPS signal
'***Variables initilized
'*** Open port, abort if recording stopped
Port = FreeFile
If Abort Then Goto ErrorHandler
On Error Resume Next
StatusMsg "Opening GPS port"
Open "Com2:" as Port nowait
If Err <> 0 Then
    ErrorMsg "Failed to open com port, err " &Err
    Goto ErrorHandler
End If
StatusMsg "GPS port opened"
SetPort Port, BAUD, NOPARITY, 8, 1, NOHANDSHAKE
StartTime = Time
TimeValid = false
Digital 1, 4, -1 'turn on the GPS if its control is connected to COM4
Do
    SetTimeout Port,35
    '*** port opened and initialized
```



APPLICATION NOTE

4

Using the Garmin GPS-16-HVS & GPS-17-HVS with the Sutron 9210 & Xpert Data Loggers

'GPS sends data out every second in the format:

```
'$GPGGA,hhmmss,llll.lll,a,nnnnn.nnn,b,t,uu,v.v,w.w,M,x.x,M,y.y,zzzz&hh<CR><LF>
>
```

```
    FlushInput Port      ' get rid of any data in the input buffer
    if Err then
        ErrorMsg "FlushInput failed " &Err
    end if
    WaitFor Port, "$GPGGA" ' waitfor message identifier
    if Err then
        ErrorMsg "Failed to get $GPGGA"
        Goto ErrorHandler
    end if
    StatusMsg "Got $GPGGA,"
    MSGID = 0
    UTC = 0
    Latitude = 0.0
    LatH = 0
    Longitude = 0.0
    LongH = 0
    GPSQual = 0
    NumSats = 0
    HorDil = 0.0
    AntAlt = 0.0
    Meters = ""
    AgeDiff = 0.0
    DiffRef = 0
    Asterisk = ""
    Cksum = 0
    strdata = ""
    Line Input port, strdata
    statusmsg strdata
    Input Port, MSGID, UTC, Latitude, LatH, Longitude, LongH, GPSQual,
    NumSats, HorDil, AntAlt, Meters, AgeDiff, DiffRef, Asterisk, Cksum
    if Err then
        ' couldn't get data
        ErrorMsg "Input error " &Err
        goto ErrorHandler
    end if
    ' we got data, now see if it is any good
    ' check to see if we have GPS lock
    statusmsg "MSG " &UTC &" " &latitude &" " &gpsqual &" " &numsats
    if (NumSats > 0) AND (GPSQual >0) then
        TimeValid = true
    end if
    loop until (TimeValid OR ((Time - StartTime)> GPSTimeout))
    if ((Time - StartTime) > GPSTimeout) then
        ErrorMsg "Timeout waiting for GPS -- NumSats: " &NumSats &" GPSQUAL: "
        &GPSQual
        goto ErrorHandler
    end if
    Hr = int(UTC / 10000)
    Min = int((UTC - (Hr * 10000))/100)
    Sec = UTC Mod 100
    GPSTime=TimeSerial(Hr, Min, Sec)
    if ((Time-GPSTime)>1) Or ((GPSTime-Time)>1) then
        Time = GPSTime
```



APPLICATION NOTE

5

Using the Garmin GPS-16-HVS & GPS-17-HVS with the Sutron 9210 & Xpert Data Loggers

```
StatusMsg "Time set to " &GPSTime
else
    StatusMsg "Time too close " &GPSTime
end if
ErrorHandler:
    Close Port
    Digital 1, 4, 0      ' turn off GPS

End Sub
```